

Bios 323 Lab X

Topic

- Stratified proportional hazards model
- Time-dependent covariates

Data Set

bmt: bone marrow transplant

group: Disease Group. 1-ALL, 2-AML Low Risk, 3-AML High Risk

t1: Time to Death or On Study Time

d1: Death Indicator (1-Dead, 0-Alive)

z2: Donor Age in Years

z7: Waiting time to transplant in days

z10: MTX Used as a Graft-Versus-Host-Prophylactic (1-Yes, 0-No)

STATA

```
. insheet using bmt.csv
. stset t1, failure(d1)
. rename z2 age
. rename z7 waiting
. rename z10 mtx
. label define grouplab 1 "ALL" 2 "AML low" 3 "AML high"
. label values group grouplab
. label define mtxlabel 1 "Yes" 0 "No"
. label values mtx mtxlabel

* fit Cox PH Model *
. stcox i.group age c.age#c.age mtx waiting, nohr
. stcox // recall the current model
. display _b[age]
. display _se[2.group]
. stcox i.group age c.age#c.age mtx waiting

* handling ties *
. stcox i.group age c.age#c.age mtx waiting, efron
. stcox i.group age c.age#c.age mtx waiting, exactm
. stcox i.group age c.age#c.age mtx waiting, exactp

* test on disease group *
* wald test *
. test 2.group 3.group
* LR test *
. estimates store mod1
. stcox age c.age#c.age mtx waiting
. estimates store mod2
. lrtest mod1 mod2, stat
```

```

* score test *
. findit scoretest_cox
. generate age2=age*age
. generate group2=0
. replace group2=1 if group==2
. generate group3=0
. replace group3=1 if group==3
. stcox group2 group3 age age2 mtx waiting
. scoretest_cox group2 group3

* test on H0: group2=group3 *
* change the reference level *
. stcox b2.group age c.age#c.age mtx waiting
. stcox i.group age c.age#c.age mtx waiting
. test 2.group==3.group
* test on interaction term *
. stcox i.group age c.age#c.age mtx waiting group#mtx
. testparm group#mtx
. test (1b.group#1.mtx) (2.group#1.mtx)

* baseline survival functions *
. stcox i.group age c.age#c.age mtx waiting
. predict h0, basechazard
. line h0 _t, c(J) sort
. predict s0, basesurv
. line s0 _t, c(J) sort

* to predict survival with specified covariates *
. generate s1=s0^exp(_b[2.group] + _b[age]*30 + _b[c.age#c.age]*900 ///
+ _b[mtx] + _b[waiting]*200)
. line s0 s1 _t, connect(J J) sort
. stcurve, survival at1(group=1, age=0, mtx=0, waiting=0) ///
at2(group=2, age=20, mtx=1, waiting=200)

* stratified analysis*
. stcox i.group age c.age#c.age waiting, strata(mtx)
. predict S0, basesurv
. generate S00=S0 if mtx==0
. generate S01=S0 if mtx==1
. line S00 S01 _t, c(J J) sort

* time varying covariate *
. stcox group2 group3 age c.age#c.age mtx waiting, tvc(group2) texp(_t>200) nohr
. stcox group2 group3 age c.age#c.age mtx, tvc(waiting) texp(exp(-0.2*_t)) nohr
. generate id=_n
. stset t1, failure(d1) id(id)
. list id _t0 _t _d mtx if id>10 & id<20, sep(0)
. stsplot new, every(365)
. list id _t0 _t _d new mtx if id>10 & id<20, sep(0)
. set seed 123
. generate xnew=int(rnormal(100, 20))
. list id _t0 _t _d new xnew if id>10 & id<20, sep(0)
. stcox i.group age c.age#c.age mtx waiting xnew

```

```

. generate start=_t0
. generate end=_t
. generate indicator=_d
. outsheet start end indicator id xnew age group waiting mtx using bmt2.csv

* fit a cox model without covariate *
. stcox, estimate

```

R

```

# stratified analysis #
> f2 <- coxph(Surv(t1, d1) ~ factor(group) + age + I(age^2)
+ waiting + strata(mtx), data=bmt)
# likelihood ratio test #
> f2a <- coxph(Surv(t1, d1) ~ factor(group) + age + I(age^2)
+ waiting, data=subset(bmt, mtx==0))
> f2b <- coxph(Surv(t1, d1) ~ factor(group) + age + I(age^2)
+ waiting, data=subset(bmt, mtx==1))
> stat <- 2*(f2a$loglik[2] + f2b$loglik[2] - f2$loglik[2])
> 1-pchisq(stat, df=5)
> f3 <- coxph(Surv(t1, d1) ~ (factor(group) + age + I(age^2)
+ waiting) * mtx, data=bmt)
> f4 <- coxph(Surv(t1, d1) ~ factor(group) + age + I(age^2)
+ waiting + mtx, data=bmt)
> anova(f3, f4) # This is not the right test #
# wald test #
> f2c <- coxph(Surv(t1, d1) ~ (group2 + group3 + age + I(age^2) + waiting)*strata(mtx), data=bmt)
> stat <- t(f2c$coef[6:10]) %*% solve(f2c$var[6:10, 6:10]) %*% f2c$coef[6:10]
> 1-pchisq(stat, df=5)

# time varying covariate #
> bmt$group2 <- ifelse(bmt$group==2, 1, 0)
> bmt$group3 <- ifelse(bmt$group==3, 1, 0)
> f5 <- coxph(Surv(t1, d1) ~ tt(group2) + group3 + age + I(age^2)
+ mtx + waiting, data=bmt, tt=function(x,t,...) x+x*I(t>200))
> f6 <- coxph(Surv(t1, d1) ~ group2 + group3 + age + I(age^2)
+ mtx + tt(waiting), data=bmt, tt=function(x,t,...) x*exp(-0.2*t))

# test that the hazard rates of the three disease groups are proportional #
> f7 <- coxph(Surv(t1, d1) ~ group2 + tt(group2) + group3 + tt(group3) + age + I(age^2) + waiting + mtx,
+ data=bmt, tt=function(x,t,...) x*log(t))
> stat <- t(f7$coef[c(2,4)]) %*% solve(f7$var[c(2,4), c(2,4)]) %*% f7$coef[c(2,4)]
> 1-pchisq(stat, df=2)

> bmt2 <- read.csv("bmt2.csv", sep="\t")
> f8 <- coxph(Surv(start, end, indicator) ~ group + age + I(age^2)
+ mtx + waiting, data=bmt2)

```